

NLPINF: A Fortran Implementation of an SQP Algorithm for Maximum-Norm Optimization Problems - User's Guide -

Address: Prof. K. Schittkowski
Siedlerstr. 3
D - 95488 Eckersdorf
Germany

Phone: (+49) 921 32887

E-mail: klaus@schittkowski.de

Web: <http://www.klaus-schittkowski.de>

Date: April, 2013

Abstract

The Fortran subroutine NLPINF solves constrained min-max or L_∞ nonlinear programming problems, where the maximum of absolute nonlinear function values is to be minimized. It is assumed that all functions are continuously differentiable. By introducing one additional variable and nonlinear inequality constraints, the problem is transformed into a general smooth nonlinear program subsequently solved by the sequential quadratic programming (SQP) code NLPQLP. An important application is data fitting, where the distance of experimental data from a model function evaluated at given experimental times is minimized by the L_∞ or maximum norm, respectively. The usage of the code is documented, and two illustrative examples are presented.

Keywords: min-max optimization, L_∞ norm, data fitting, approximation, SQP, sequential quadratic programming, nonlinear programming, numerical algorithms, Fortran codes

1 Introduction

Min-max optimization problems arise in many practical situations, for example in approximation or when fitting a model function to given data in the L_∞ -norm. In this particular case, a mathematical model is available in form of one or several equations, and the goal is to estimate some unknown parameters of the model. Exploited are available experimental data, to minimize the distance of the model function, in most cases evaluated at certain time values, from measured data at the same time values. An extensive discussion of data fitting especially in case of dynamical systems is given by Schittkowski [8], and the code NLPINF is part of the software system EASY-FIT.

The mathematical problem we want to solve, is given in the form

$$\begin{aligned} & \min \max\{|f_i(x)|, i = 1, \dots, l\} \\ x \in \mathbb{R}^n : & \quad g_j(x) = 0, \quad j = 1, \dots, m_e, \\ & \quad g_j(x) \geq 0, \quad j = m_e + 1, \dots, m, \\ & \quad x_l \leq x \leq x_u. \end{aligned} \tag{1}$$

It is assumed that f_1, \dots, f_l and g_1, \dots, g_m are continuously differentiable functions.

Some test examples for L_∞ -approximation are studied in Schittkowski [13, 15], where the number of data points is extremely large, i.e., up to 60,000,000. An active set strategy is applied to reduce the size of the Jacobian matrix of the quadratic programming subproblem, see Schittkowski [7] for details.

In this paper, we consider the question how an existing nonlinear programming code can be used to solve constrained min-max problems in an efficient and robust way after a suitable transformation. In a very similar way, also L_1 and least squares problems can be solved efficiently by an SQP code, see Schittkowski [8, 11, 14, 16].

The transformation of an L_∞ problem into a special nonlinear program is described in Section 2. Sections 3 to 5 contains a complete documentation of the Fortran code and two example implementations.

2 The Transformed Optimization Problem

We consider the constrained nonlinear min-max or L_∞ problem (1), and introduce an additional variables z and $2l$ additional nonlinear equality constraints of the form

$$\begin{aligned} f_i(x) + z & \geq 0, \\ -f_i(x) + z & \geq 0, \end{aligned} \tag{2}$$

$i = 1, \dots, l$. The following equivalent problem is to be solved by an SQP method,

$$\begin{aligned}
& \min z \\
& g_j(x) = 0 \quad , \quad j = 1, \dots, m_e \quad , \\
& g_j(x) \geq 0 \quad , \quad j = m_e + 1, \dots, m \quad , \\
(x, z) \in \mathbb{R}^{n+1} : & f_i(x) + z \geq 0 \quad , \quad i = 1, \dots, l \quad , \\
& -f_i(x) + z \geq 0 \quad , \quad i = 1, \dots, l \quad , \\
& x_l \leq x \leq x_n \quad , \\
& z \geq 0 \quad .
\end{aligned} \tag{3}$$

In this case, the quadratic programming subproblem which has to be solved in each step of an SQP method, has the form

$$\begin{aligned}
& \min \frac{1}{2}(d^T, e)B_k \begin{pmatrix} d \\ e \end{pmatrix} + e \\
& \nabla g_j(x_k)^T d + g_j(x_k) = 0 \quad , \quad j = 1, \dots, m_e \quad , \\
& \nabla g_j(x_k)^T d + g_j(x_k) \geq 0 \quad , \quad j = m_e + 1, \dots, m \quad , \\
(d, e) \in \mathbb{R}^{n+1} : & \nabla f_i(x_k)^T d + e + f_i(x_k) + z_k \geq 0 \quad , \quad i = 1, \dots, l \quad , \\
& -\nabla f_i(x_k)^T d + e - f_i(x_k) + z_k \geq 0 \quad , \quad i = 1, \dots, l \quad , \\
& x_l - x_k \leq d \leq x_u - x_k \quad , \\
& e \geq 0 \quad .
\end{aligned} \tag{4}$$

$B_k \in \mathbb{R}^{n+1} \times \mathbb{R}^{n+1}$ is a quasi-Newton update matrix of the Lagrangian function of (3). A new iterate is then obtained from

$$x_{k+1} = x_k + \alpha_k d_k \quad , \quad z_{k+1} = z_k + \alpha_k e_k \quad ,$$

where $d_k \in \mathbb{R}^n$ and $e_k \in \mathbb{R}$ are a solution of (4) and α_k a steplength parameter obtained from forcing a sufficient descent of a merit function.

The proposed transformation (3) is independent of the SQP method used, so that available codes can be used in the form of a *black box*. However, an active set strategy is recommended to reduce the number of constraints, if l becomes large, e.g., the code NLPQLB [13].

A final remark concerns the theoretical convergence of the algorithm. Since the original problem is transformed into a general nonlinear programming problem, we can apply all convergence results known for SQP methods. If an augmented Lagrangian function is preferred for the merit function, a global convergence theorem is found in Schittkowski [3], see also [7] for convergence of the active set strategy. The theorem states that when starting from an arbitrary initial value, a Karush-Kuhn-Tucker point

is approximated, i.e., a point satisfying the necessary optimality conditions. If, on the other hand, an iterate is sufficiently close to an optimal solution and if the steplength is 1, then the convergence speed of the algorithm is superlinear, see Powell [1] for example. This remark explains the fast final convergence rate one observes in practice.

3 Calling Sequence

In this section, we describe the arguments of subroutine NLPINF in detail.

Usage:

```

CALL NLPINF (      M,      ME,  LMMAX,      L,      N,
/              NMAX,  LMNN2,      X,  FUNC,      RES,
/              GRAD,      U,      XL,      XU,      ACC,
/              ACCQP,  RESSIZ,  MAXFUN  MAXIT,      MAXNM,
/              RHOB,  IPRINT,      IOUT,      IFAIL,      WA,
/              LWA,      KWA,      LKWA,  LOGWA,  LLOGWA )

```

Definition of the parameters:

- M : Number of constraints, i.e., m .
- ME : Number of equality constraints, i.e., m_e .
- LMMAX : Row dimension of GRAD and dimension of FUNC. LMMAX must be at least one and not smaller than $L + M$.
- L : Number of terms in objective function, i.e., l .
- N : Number of variables, i.e., n .
- NMAX : Dimensioning parameter, at least two and greater than $N + 1$.
- LMNN2 : Dimensioning parameter, must be set to $M + 2*N + 2*L + 4$ when calling NLPINF.
- X(NMAX) : On input, the first N positions of X have to contain an initial guess for the solution. On return, X is replaced by the last computed iterate.

FUNC(LMMAX) : Function values passed to NLPINF by reverse communication, i.e., the first L positions contain the L residual values $f_i(x)$, $i = 1, \dots, l$, the subsequent M coefficients the constraint values $g_j(x)$, $j = 1, \dots, m$.

RES : On return, RES contains the objective function value $\max\{|f_1(x)|, \dots, |f_l(x)|\}$.

GRAD(LMMAX, LNMAX) : The array is used to pass gradients of residuals and constraints to NLPINF by reverse communication. In the driving program, the row dimension of GRAD must be equal to LM-MAX. The first L rows contain L gradients of residual functions $\nabla f_i(x)$ at x , $i = 1, \dots, l$, the subsequent M rows gradients of constraint functions $\nabla g_j(x)$, $j = 1, \dots, m$.

U(LMNN2) : On return, U contains the multipliers with respect to the last computed iterate. The first M locations contain the multipliers of the M nonlinear constraints, the subsequent N locations the multipliers of the lower bounds, and the following N locations the multipliers of the upper bounds. At an optimal solution, all multipliers with respect to inequality constraints should be nonnegative.

XL(NMAX), XU(NMAX) : On input, the one-dimensional arrays XL and XU must contain the upper and lower bounds of the variables.

ACC : The user has to specify the desired final accuracy (e.g. 1.0D-7). The termination accuracy should not be much smaller than the accuracy by which gradients are computed.

ACCQP : The tolerance is passed to the QP solver to perform several tests, for example whether optimality conditions are satisfied or whether a number is considered as zero or not. If ACCQP is less or equal to zero, then the machine precision is computed by NLPQLP and subsequently multiplied by 1.0D+4.

RESSIZE : The user must indicate a guess for the approximate size of the objective function. RESSIZE must not be negative.

MAXFUN : The integer variable defines an upper bound for the number of function calls during the line search (e.g. 20).

MAXIT : Maximum number of outer iterations, where one iteration corresponds to one formulation and solution of the quadratic programming subproblem, or, alternatively, one evaluation of gradients (e.g. 100).

MAXNM : Stack size for storing merit function values at previous iterations for non-monotone line search (e.g. 10).

RHOB : Parameter for initializing a restart in case of IFAIL=2 by setting the BFGS-update matrix to $\text{rhob} \cdot \mathbf{I}$, where \mathbf{I} denotes the identity matrix. The number of restarts is bounded by MAXFUN. No restart is performed if RHOB is set to zero. Must be non-negative (e.g. 100).

IPRINT : Specification of the desired output level:

- 0 - No output of the program.
- 1 - Only final convergence analysis.
- 2 - One line of intermediate results for each iteration.
- 3 - More detailed information for each iteration.
- 4 - More line search data displayed.

IOUT : Integer indicating the desired output unit number, i.e., all write-statements start with 'WRITE(IOUT,...)'.

IFAIL : The parameter shows the reason for terminating a solution process. Initially IFAIL must be set to zero. On return IFAIL could contain the following values:

- 2 - Compute new gradient values.
- 1 - Compute new function values.
- 0 - Optimality conditions satisfied.
- 1 - Stop after MAXIT iterations.
- 2 - Uphill search direction.
- 3 - Underflow when computing new BFGS-update matrix.
- 4 - Line search exceeded MAXFUN iterations.
- 5 - Length of a working array too short.
- 6 - False dimensions, $M > MMAX$, $N \geq NMAX$, or $MNN2 \neq M + N + N + 2$.
- 7 - Search direction close to zero at infeasible iterate.
- 8 - Starting point violates lower or upper bound.
- 9 - Wrong input parameter, e.g., MODE, IPRINT, IOUT.
- 10 - Inconsistency in QP, division by zero.
- >100 - Error message of QP solver.

WA(LWA) : WA is a real working array of length LWA.

LWA : Length of the real working array WA. LWA must be at least

$5*N*N/2 + (L+L+M)*N + 24*L + 12*M + 45*N + 250.$

KWA(LKWA) : KWA is an integer working array of length LKWA.
 LKWA : Length of the integer working array KWA. LKWA must be at least $N + 28$. On return, KWA(1) and KWA(2) contain the number of function and derivative evaluations, respectively.

LOGWA(LLOGWA) : Logical working array of length LLOGWA.
 LLOGWA : Length of the logical array LOGWA. The length LLOGWA of the logical array must be at least $4*L + 2*M + 10$.

4 Program Organization

All declarations of real numbers must be done in double precision. Subroutine NLPINF must be linked with the user-provided main program, the SQP code NLPQLP [12], and the quadratic programming code QL [10].

NLPINF is implemented in form of a Fortran subroutine. Model functions and gradients are passed by reverse communication. The user has to provide functions and gradients in the same program which executes NLPINF, according to the following rules:

1. Choose starting values for the variables to be optimized, and store them in the first N positions of X .
2. Compute residual and constraint function values values, and store them in a one-dimensional double precision array FUNC. The first L positions contain the L residual values $f_i(x)$, $i = 1, \dots, l$, the subsequent M coefficients the constraint values $g_j(x)$, $j = 1, \dots, m$.
3. Compute gradients of residual and constraint functions, and store them in a two-dimensional double precision array GRAD. The first L rows contain gradients of residual functions $\nabla f_i(x)$ at x , $i = 1, \dots, l$, the subsequent M rows gradients of constraint functions $\nabla g_j(x)$, $j = 1, \dots, m$.
4. Set IFAIL=0 and execute NLPINF.
5. If NLPINF returns with IFAIL=-1, compute residual function values and constraint values for the arguments found in X , and store them in FUNC in the order shown above. Then call NLPINF again, but do not change IFAIL.
6. If NLPINF terminates with IFAIL=-2, compute gradient values subject to variables stored in X , and store them in GRAD as indicated above. Then call NLPINF again without changing IFAIL.
7. If NLPINF terminates with IFAIL=0, the internal stopping criteria are satisfied. The variable values found in X are considered as a local solution of the min-max optimization problem.

8. In case of $IFAIL > 0$, an error occurred.

If analytical derivatives are not available, additional function calls are required for gradient approximations, for example by forward differences, two-sided differences, or even higher order formulae.

Some of the termination reasons depend on the accuracy used for approximating gradients. If we assume that all functions and gradients are computed within machine precision and that the implementation is correct, there remain only the following possibilities that could cause an error message:

1. The termination parameter ACC is too small, so that the numerical algorithm plays around with round-off errors without being able to improve the solution. Especially the Hessian approximation of the Lagrangian function becomes unstable in this case. A straightforward remedy is to restart the optimization cycle again with a larger stopping tolerance.
2. The constraints are contradicting, i.e., the set of feasible solutions is empty. There is no way to find out, whether a general nonlinear and non-convex set possesses a feasible point or not. Thus, the nonlinear programming algorithms will proceed until running in any of the mentioned error situations. In this case, there the correctness of the model must be checked very carefully.
3. Constraints are feasible, but some of them there are degenerate, for example if some of the constraints are redundant. One should know that SQP algorithms require satisfaction of the so-called constraint qualification, i.e., that gradients of active constraints are linearly independent at each iterate and in a neighborhood of the optimal solution. In this situation, it is recommended to check the formulation of the model.

However, some of the error situations do also occur, if because of wrong or non-accurate gradients, the quadratic programming subproblem does not yield a descent direction for the underlying merit function. In this case, one should try to improve the accuracy of function evaluations, scale the model functions in a proper way, or start the algorithm from other initial values.

5 Examples

To give a simple example how to organize the code in case of two explicitly given functions, we consider Rosenbrock's banana function, see test problem TP1 of Hock and Schittkowski [2],

$$x_1, x_2 \in \mathbb{R} : \min \max\{|10(x_2 - x_1^2)|, |1 - x_1|\} \quad (5)$$

The Fortran source code for executing NLPINF is listed below. Gradients are computed analytically.


```

      IMPLICIT      NONE
      INTEGER      N, M, ME, L, LMNN2, LWA, LKWA, LLOGWA
      PARAMETER    (N = 4, M = 0, ME = 0, L = 2)
      PARAMETER    (LMNN2 = M + 2*N + 3*L + 2,
/                 LWA   = 5*N*N/2 + (L+L+M)*N + 24*L + 12*M
/                 + 40*N + 250,
/                 LKWA  = N + 28,
/                 LLOGWA = 4*L + 2*M + 10)
      INTEGER      MAXFUN, MAXIT, IPRINT, MAXNM, IOUT, IFAIL,
/                 KWA(LKWA)
      DOUBLE PRECISION RES, ACC, ACCQP, RESSIZ, RHOB, EPS,
/                 X(N+2), FUNC(L+M), GRAD(L+M,N),
/                 U(LMNN2), XL(N+2), XU(N+2), WA(LWA)
      LOGICAL      LOGWA(LLOGWA)

C
C set parameters
C
      ACC   = 1.0D-14
      ACCQP = ACC
      RESSIZ = 0.0D0
      RHOB  = 0.0D0
      MAXFUN = 20
      MAXIT  = 100
      MAXNM  = 0
      IPRINT = 2
      IOUT   = 6
      IFAIL  = 0

C
C starting values and bounds
C
      X(1)  = -1.2D0
      XL(1) = -1.0D5
      XU(1) =  1.0D5
      X(2)  =  1.0D0
      XL(2) = -1.0D5
      XU(2) =  1.0D5

C
C execute NLPINF by reverse communication
C
1 CONTINUE
  IF ((IFAIL.EQ.0).OR.(IFAIL.EQ.-1)) THEN
    FUNC(1) = 10.0D0*(X(2) - X(1)**2)
    FUNC(2) = 1.0D0 - X(1)
  ENDIF
  IF ((IFAIL.EQ.0).OR.(IFAIL.EQ.-2)) THEN
    GRAD(1,1) = -20.0D0*X(1)

```

```

        GRAD(1,2) = 10.0D0
        GRAD(2,1) = -1.0D0
        GRAD(2,2) = 0.0D0
    ENDIF
C
C call NLPINF
C
    CALL NLPINF(M, ME, L+M, L, N, N+2, LMNN2, X, FUNC, RES,
/           GRAD, U, XL, XU, ACC, ACCQP, RESSIZ, MAXFUN, MAXIT,
/           MAXNM, RHOB, IPRINT, IOUT, IFAIL, WA, LWA, KWA,
/           LKWA, LOGWA, LLOGWA)
    IF (IFAIL.LT.0) GOTO 1
C
    STOP
    END

```

The following output appears on the screen:

```

-----
START OF THE SEQUENTIAL QUADRATIC PROGRAMMING ALGORITHM
-----

```

Parameters:

```

N      =      5
M      =      4
ME     =      0
MODE   =      0
ACC    =  0.1000D-13
ACCQP  =  0.1000D-13
STPMIN =  0.1000D-13
MAXFUN =      20
MAXNM  =      0
MAXIT  =     100
IPRINT =      2

```

Output in the following order:

```

IT     - iteration number
F      - objective function value
SCV    - sum of constraint violations
NA     - number of active constraints
I      - number of line search iterations
ALPHA  - steplength parameter
DELTA  - additional variable to prevent inconsistency
KKT    - Karush-Kuhn-Tucker optimality criterion

```

IT	F	SCV	NA	I	ALPHA	DELTA	KKT
1	0.00000000D+00	0.66D+01	4	0	0.00D+00	0.00D+00	0.75D+01
2	0.16121212D+01	0.23D+00	4	1	0.10D+01	0.00D+00	0.33D+00
3	0.12941070D+01	0.00D+00	4	1	0.10D+01	0.00D+00	0.13D+01
4	0.67818712D+00	0.30D+01	4	2	0.48D+00	0.00D+00	0.67D+00
5	0.25925355D+00	0.18D+01	4	1	0.10D+01	0.00D+00	0.41D+00
6	0.00000000D+00	0.67D+00	4	1	0.10D+01	0.00D+00	0.13D-03
7	0.19259299D-33	0.00D+00	4	1	0.10D+01	0.00D+00	0.11D-30

--- Final Convergence Analysis at Last Iterate ---

Objective function value: F(X) = 0.19259299D-33
Solution values: X =
0.10000000D+01 0.10000000D+01 0.00000000D+00 0.00000000D+00
0.19259299D-33
Multiplier values: U =
0.00000000D+00 0.50000000D+00 0.00000000D+00 0.50000000D+00
0.00000000D+00 0.00000000D+00 0.00000000D+00 0.00000000D+00
0.00000000D+00 0.00000000D+00 0.00000000D+00 0.00000000D+00
0.00000000D+00 0.00000000D+00
Constraint values: G(X) =
0.19259299D-33 0.19259299D-33 0.19259299D-33 0.19259299D-33
Distance from lower bound: XL-X =
-0.10000100D+06 -0.10000100D+06 0.00000000D+00 0.00000000D+00
-0.10000000D+31
Distance from upper bound: XU-X =
0.99999000D+05 0.99999000D+05 0.00000000D+00 0.00000000D+00
0.10000000D+31
Number of function calls: NFUNC = 8
Number of gradient calls: NGRAD = 7
Number of calls of QP solver: NQL = 7

--- Final Convergence Analysis of NLPINF ---

Maximum function value: RES = 0.19259299D-33
Function values: F(X) =
0.00000000D+00 0.00000000D+00
Solution: X =
0.10000000D+01 0.10000000D+01 0.00000000D+00 0.00000000D+00
Multiplier values: U =
0.00000000D+00 0.00000000D+00 0.00000000D+00 0.00000000D+00
0.00000000D+00 0.00000000D+00 0.00000000D+00 0.00000000D+00
Number of function calls: NFUNC = 8
Number of derivative calls: NGRAD = 7

Another example illustrates data fitting in the L_∞ norm by NLPINF. The time-dependent model function is

$$h(x, t) = \frac{x_1 t(t + x_2)}{t^2 + x_3 t + x_4} , \quad (6)$$

i.e., $n = 4$ and $x = (x_1, \dots, x_4)^T$. The goal is to fit $l = 11$ given experimental data

i	1	2	3	4	5	6	7	8	9	10	11
t_i	0.0625	0.0714	0.0823	0.1	0.125	0.167	0.25	0.5	1	2	4
y_i	0.0246	0.0235	0.0323	0.0342	0.0456	0.0627	0.0844	0.16	0.1735	0.1947	0.1957

under two additional equality constraints, to hit the first and last data point exactly,

$$\begin{aligned} g_1(x) &= h(x, t_1) - y_1 = 0 , \\ g_2(x) &= h(x, t_l) - y_l = 0 . \end{aligned} \quad (7)$$

The resulting optimization problem is

$$\begin{aligned} & \min \max\{|h(x, t_i) - y_i|, i = 1, \dots, l\} \\ x \in \mathbb{R}^n : & \quad g_1(x) = 0 , \\ & \quad g_2(x) = 0 , \\ & \quad 0 \leq x . \end{aligned} \quad (8)$$

The corresponding code and the displayed output of NLPINF are listed below.

```

IMPLICIT          NONE
INTEGER          N, M, ME, L, LMNN2, LWA, LKWA, LLOGWA
PARAMETER       (N = 4, M = 2, ME = 2, L = 11)
PARAMETER       (LMNN2 = M + 2*L + 2*N + 4,
/               LWA    = 5*N*N/2 + (L+L+M)*N + 24*L + 12*M
/               + 40*N + 250,
/               LKWA   = N + 28,
/               LLOGWA = 4*L + 2*M + 10)
INTEGER          MAXFUN, MAXIT, IPRINT, MAXNM, IOUT, IFAIL,
/               KWA(LKWA), I, J
DOUBLE PRECISION RES, ACC, ACCQP, RESSIZ, RHOB, EPS,
/               X(N+2), FUNC(L+M), GRAD(L+M,N), U(LMNN2),
/               XL(N+2), XU(N+2), WA(LWA), T(L), Y(L), W(N)
LOGICAL          LOGWA(LLOGWA)
DATA            T/0.0625D0,0.0714D0,0.0823D0,0.1000D0,0.1250D0,
/               0.1670D0,0.2500D0,0.5000D0,1.0000D0,2.0000D0,
/               4.0000D0/
DATA            Y/0.0246D0,0.0235D0,0.0323D0,0.0342D0,0.0456D0,
/               0.0627D0,0.0844D0,0.1600D0,0.1735D0,0.1947D0,
/               0.1957D0/

```

C

```

C   set parameters
C
    ACC   = 1.0D-14
    ACCQP = 1.0D-14
    RESSIZ = 1.0D0
    RHOB  = 0.0D0
    MAXFUN = 20
    MAXIT  = 100
    MAXNM  = 20
    IPRINT = 2
    IOUT   = 6
    IFAIL  = 0

C
C   starting values and bounds
C
    X(1) = 0.25D0
    X(2) = 0.39D0
    X(3) = 0.415D0
    X(4) = 0.39D0
    DO I = 1,N
        XL(I) = 0.0D0
        XU(I) = 1.0D5
    ENDDO

C
C   execute NLPINF by reverse communication
C
1 CONTINUE
  IF ((IFAIL.EQ.0).OR.(IFAIL.EQ.-1)) THEN
    DO J = 1,L
      CALL H(T(J), Y(J), N, X, FUNC(J))
    ENDDO
    CALL H(T(1), Y(1), N, X, FUNC(L+1))
    CALL H(T(L), Y(L), N, X, FUNC(L+2))
  ENDIF
  IF ((IFAIL.EQ.0).OR.(IFAIL.EQ.-2)) THEN
    DO J = 1,L
      CALL DH(T(J), N, X, W)
      DO I=1,N
        GRAD(J,I) = W(I)
      ENDDO
    ENDDO
    CALL DH(T(1), N, X, W)
    DO I=1,N
      GRAD(L+1,I) = W(I)
    ENDDO
    CALL DH(T(L), N, X, W)
  
```

```

        DO I=1,N
            GRAD(L+2,I) = W(I)
        ENDDO
    ENDIF
C
C call NLPINF
C
    CALL NLPINF(M, ME, L+M, L, N, N+2, LMNN2, X, FUNC, RES,
/           GRAD, U, XL, XU, ACC, ACCQP, RESSIZ, MAXFUN, MAXIT,
/           MAXNM, RHOB, IPRINT, IOUT, IFAIL, WA, LWA, KWA,
/           LKWA, LOGWA, LLOGWA)
    IF (IFAIL.LT.0) GOTO 1
C
    STOP
    END
C
C data fitting function
C
    SUBROUTINE      H(T, Y, N ,X, F)
    IMPLICIT        NONE
    INTEGER         N
    DOUBLE PRECISION T, Y, X(N), F
C
    F = X(1)*T*(T + X(2))/(T**2 + X(3)*T + X(4)) - Y
C
    RETURN
    END
C
C partial derivatives
C
    SUBROUTINE      DH(T, N ,X, DF)
    IMPLICIT        NONE
    INTEGER         N
    DOUBLE PRECISION T, X(N), DF(N)
C
    DF(1) = T*(T + X(2))/(T**2 + X(3)*T + X(4))
    DF(2) = X(1)*T/(T**2 + X(3)*T + X(4))
    DF(3) = -X(1)*T**2*(T + X(2))/(T**2 + X(3)*T + X(4))**2
    DF(4) = -X(1)*T*(T + X(2))/(T**2 + X(3)*T + X(4))**2
C
    RETURN
    END

```

START OF THE SEQUENTIAL QUADRATIC PROGRAMMING ALGORITHM

Parameters:

N = 5
M = 24
ME = 2
MODE = 0
ACC = 0.1000D-13
ACCQP = 0.1000D-13
STPMIN = 0.1000D-13
MAXFUN = 20
MAXNM = 20
MAXIT = 100
IPRINT = 2

Output in the following order:

IT - iteration number
F - objective function value
SCV - sum of constraint violations
NA - number of active constraints
I - number of line search iterations
ALPHA - steplength parameter
DELTA - additional variable to prevent inconsistency
KKT - Karush-Kuhn-Tucker optimality criterion

IT	F	SCV	NA	I	ALPHA	DELTA	KKT
1	0.10000000D+01	0.55D-01	24	0	0.00D+00	0.00D+00	0.10D+01
2	0.14273413D-01	0.76D-02	3	1	0.10D+01	0.00D+00	0.12D-01
3	0.15957258D-01	0.38D-03	3	1	0.10D+01	0.00D+00	0.73D-02
4	0.12379500D-01	0.11D-03	4	1	0.10D+01	0.00D+00	0.12D-03
5	0.12346540D-01	0.16D-05	4	1	0.10D+01	0.00D+00	0.23D-03
6	0.12117156D-01	0.18D-04	4	1	0.10D+01	0.00D+00	0.13D-02
7	0.10853183D-01	0.58D-03	4	1	0.10D+01	0.00D+00	0.68D-03
8	0.10681358D-01	0.17D-05	5	1	0.10D+01	0.00D+00	0.79D-06
9	0.10681368D-01	0.45D-11	5	1	0.10D+01	0.00D+00	0.34D-11
10	0.10681368D-01	0.49D-16	5	1	0.10D+01	0.00D+00	0.37D-16

--- Final Convergence Analysis at Best Iterate ---

Best result at iteration: ITER = 10
Objective function value: F(X) = 0.10681368D-01
Solution values: X =
0.19191422D+00 0.36223614D+00 0.23065604D+00 0.18877354D+00
0.10681368D-01
Multiplier values: U =

```

0.28276724D+00 0.58912518D-01 0.00000000D+00 0.00000000D+00
0.00000000D+00 0.00000000D+00 0.00000000D+00 0.00000000D+00
0.00000000D+00 0.40030440D+00 0.00000000D+00 0.00000000D+00
0.00000000D+00 0.00000000D+00 0.00000000D+00 0.00000000D+00
0.00000000D+00 0.00000000D+00 0.00000000D+00 0.36133107D+00
0.00000000D+00 0.23836453D+00 0.00000000D+00 0.00000000D+00
0.00000000D+00 0.00000000D+00 0.00000000D+00 0.00000000D+00
0.00000000D+00 0.00000000D+00 0.00000000D+00 0.00000000D+00
0.00000000D+00 0.00000000D+00
Constraint values:          G(X) =
0.34694470D-17 0.27755576D-16 0.10681368D-01 0.15430705D-01
0.11109890D-01 0.16469661D-01 0.15196771D-01 0.14450944D-01
0.21362736D-01 0.39898640D-16 0.21362736D-01 0.10965605D-01
0.10681368D-01 0.10681368D-01 0.59320313D-02 0.10252847D-01
0.48930750D-02 0.61659648D-02 0.69117920D-02 -0.17347235D-17
0.21362736D-01 -0.15612511D-16 0.10397131D-01 0.10681368D-01
Distance from lower bound:  XL-X =
-0.19191422D+00 -0.36223614D+00 -0.23065604D+00 -0.18877354D+00
-0.10000000D+31
Distance from upper bound:  XU-X =
0.99999808D+05 0.99999638D+05 0.99999769D+05 0.99999811D+05
0.10000000D+31
Number of function calls:   NFUNC =      10
Number of gradient calls:  NGRAD =      10
Number of calls of QP solver: NQL   =      10

```

--- Final Convergence Analysis of NLPINF ---

```

Maximum function value:    RES   = 0.10681368D-01
Function values:          F(X) =
0.34694470D-17 0.47493368D-02 0.42852147D-03 0.57882931D-02
0.45154033D-02 0.37695761D-02 0.10681368D-01 -0.10681368D-01
0.10681368D-01 0.28423718D-03 0.27755576D-16
Solution:                  X =
0.19191422D+00 0.36223614D+00 0.23065604D+00 0.18877354D+00
Multiplier values:        U =
0.28276724D+00 0.58912518D-01 0.00000000D+00 0.00000000D+00
0.00000000D+00 0.00000000D+00 0.00000000D+00 0.00000000D+00
0.00000000D+00 0.00000000D+00
Constraint values:          G(X) =
0.34694470D-17 0.27755576D-16
Number of function calls:  NFUNC =     10
Number of derivative calls: NGRAD =     10

```


References

- [1] Powell M.J.D. (1978): *A fast algorithm for nonlinearly constraint optimization calculations*, in: Numerical Analysis, G.A. Watson ed., Lecture Notes in Mathematics, Vol. 630, Springer, Berlin
- [2] Hock W., Schittkowski K. (1981): *Test Examples for Nonlinear Programming Codes*, Lecture Notes in Economics and Mathematical Systems, Vol. 187, Springer
- [3] Schittkowski K. (1983): *On the convergence of a sequential quadratic programming method with an augmented Lagrangian search direction*, Optimization, Vol. 14, 197-216
- [4] Schittkowski K. (1985/86): *NLPQL: A Fortran subroutine solving constrained nonlinear programming problems*, Annals of Operations Research, Vol. 5, 485-500
- [5] Schittkowski K. (1987a): *More Test Examples for Nonlinear Programming*, Lecture Notes in Economics and Mathematical Systems, Vol. 182, Springer
- [6] Schittkowski K. (1988): *Solving nonlinear least squares problems by a general purpose SQP-method*, in: Trends in Mathematical Optimization, K.-H. Hoffmann, J.-B. Hiriart-Urruty, C. Lemarechal, J. Zowe eds., International Series of Numerical Mathematics, Vol. 84, Birkhäuser, 295-309
- [7] Schittkowski K. (1992): *Solving nonlinear programming problems with very many constraints*, Optimization, Vol. 25, 179-196
- [8] Schittkowski K. (2002): *Numerical Data Fitting in Dynamical Systems*, Kluwer Academic Publishers, Dordrecht
- [9] Schittkowski K. (2002): *EASY-FIT: A software system for data fitting in dynamic systems*, Structural and Multidisciplinary Optimization, Vol. 23, No. 2, 153-169
- [10] Schittkowski K. (2003): *QL: A Fortran code for convex quadratic programming - User's guide*, Report, Department of Mathematics, University of Bayreuth
- [11] Schittkowski K. (2005): *DFNLP: A Fortran Implementation of an SQP-Gauss-Newton Algorithm - User's Guide, Version 2.0*, Report, Department of Computer Science, University of Bayreuth
- [12] Schittkowski K. (2006): *NLPQLP: A Fortran implementation of a sequential quadratic programming algorithm with distributed and non-monotone line search - user's guide, version 2.2*, Report, Department of Computer Science, University of Bayreuth

- [13] Schittkowski K. (2007): *NLPQLB: A Fortran implementation of a sequential quadratic programming algorithm with active set strategy for solving optimization problems with a very large number of nonlinear constraints - user's guide, version 2.0*, Report, Department of Computer Science, University of Bayreuth
- [14] Schittkowski K. (2007): *NLPLSQ: A Fortran implementation of an SQP-Gauss-Newton algorithm for least-squares optimization - user's guide*, Report, Department of Computer Science, University of Bayreuth
- [15] Schittkowski K. (2007): *An active set strategy for solving optimization problems with up to 60,000,000 nonlinear constraints*, submitted for publication
- [16] Schittkowski K. (2008): *NLPL1: A Fortran implementation of an SQP-Gauss-Newton algorithm for minimizing sums of absolute function values - user's guide*, Report, Department of Computer Science, University of Bayreuth